

# QSYer für den YAESU FT-897 und FT-857

28. September 2018  
Matthias DD1US

Hallo,

schon vor einigen Jahren hat ein befreundeter OM, Peter Horlacher mit dem Rufzeichen DH1PAX, eine Frequenzeingabe-Tastatur für den FT-817 entwickelt. Diese Tastatur wird am rückseitigen ACC-Anschluss des FT-817 angeschlossen. Sie kann ebenso am CAT-Anschluss des FT897 und des FT857 betrieben werden. Sie erlaubt die direkte Eingabe einer Frequenz über eine 4x3 Matrixtastatur. Hier die Funktionsbeschreibung der Tastatur:

Die Frequenzeingabe erfolgt über das Tastenfeld, dabei ist die Frequenz 8-stellig einzugeben.

Beispiele:

Für 438,125 MHz ist 43812500 einzugeben oder man verkürzt mit der Raute #, d.h. 438125# geht auch.

Für 145,500 MHz ist 14550000 einzugeben oder verkürzt 1455#.


Für 28,500 MHz ist 02850000 einzugeben oder verkürzt 0285# .

Für 3,685 MHz ist 00368500 einzugeben oder verkürzt 003685#.


Die Tastatur programmiert übrigens automatisch die korrekte Relaisablage im 10m, 2m und 70cm Band wenn Frequenzen im Relaisbereich dieser Bänder eingegeben werden.

Da ich an meinem FT-897 bereits einen Antennentuner und ein PC-Interface am CAT-Port angeschlossen habe, hat mir Peter empfohlen das serielle Interface am Mikrofonanschluss zu verwenden. Dies ist übrigens nur bei den YAESU FT-897 und FT-857 Geräten möglich, der FT-817 verfügt nicht über diese Möglichkeit. Dieses Interface kann im Einstellmenü des Funkgerätes unter dem Menüpunkt 59 „Mic Select“ aktiviert werden und kann dann parallel zu dem CAT-Interface auf der Rückseite des Gerätes verwendet werden.

Im Normalfall ist die Funktion „NOR“ ausgewählt und die Belegung des 8 poligen RJ45 Steckers ist wie folgt:



Man kann aber auch die Funktion „CAT“ auswählen, dann ändert sich die Belegung des RJ-45 Steckers wie folgt:




Die Eingänge für Up und Down sind also in Signale einer seriellen Schnittstelle TXD und RXD geändert. Die anderen Pins behalten ihre Funktion. Das Protokoll entspricht genau der CAT-Schnittstelle auf der Rückseite des Funkgerätes.

Als Tastatur wird eine 3x4 Tastaturmatrix verwendet. Sie ist beispielsweise von der Firma Conrad unter der Bestellnummer 709840 für 8,39 Euro erhältlich. Die Tastatur hat auf der Rückseite eine 7 polige Stiftleiste.




Die Auswertung der Tastaturmatrix und Kommunikation mit dem Transceiver übernimmt ein 8bit uC des Typs ATMEL AVR AT90S2313P. Die Stromversorgung mit +5V DC erfolgt direkt aus der Mikrofonbuchse. Hier das Schaltbild welches gegenüber der Originalschaltung von Peter nur leicht verändert wurde (5V Spannungsregler entfällt, Leuchtdiode zur Anzeige der aktivierten Tastatur).



Ich habe die Signale UB=5V, GND, TXD und RXD über einen 4 poligen Einschalter mit dem 8 poligen Westernstecker verbunden. Über den 4 poligen Schalter kann ich somit die komplette Schaltung vom Mikrofonkabel abtrennen, wenn ich beispielsweise ein Fernsteuermikrofon wie das YAESU MH-59 verwenden möchte.

Peter DH1PAX war so nett mir einen programmierten uC sowie einige der anderen Bauteile zur Verfügung zu stellen. Die kleine Schaltung habe ich auf einer Lochrasterplatine aufgebaut und in ein Gehäuse eingebaut.

Natürlich benötigt man noch einen Anschluss, um das Mikrofon noch anschließen zu können. In meinem Fall habe ich eine 8 polige Rundbuchse verwendet die wie folgt belegt ist:



Die 8 Adern von dem 8 poligen Westernstecker sind direkt mit der Rundbuchse verbunden.

Damit kann ich auch die gängigen Standmikrofone wie das YAESU MD-1 verwenden.

Abschließend noch Bilder meines Aufbaus:





Ich möchte nochmals darauf hinweisen, dass ich hier nur kleine Modifikationen bezüglich der Originalschaltung von Peter DH1PAX vorgenommen habe. Ihn nochmals vielen Dank für das Veröffentlichen seines Konzeptes inklusive des Sourcecodes und seiner freundlichen Unterstützung.

Ich konnte die Tastatur noch nicht testen, da mein FT-897 an einem anderen QTH steht. Ich werde dies aber baldmöglichst nachholen, worauf ich mich schon freue.

Ich freue mich stets über Rückmeldungen.

Im Voraus vielen Dank und viele Grüße

Matthias DD1US

Email: [DDIUS@AMSAT.ORG](mailto:DDIUS@AMSAT.ORG)

Homepage: <http://www.dd1us.de>

Anhang: BASCOM Source Code von Peter DH1PAX

Anhang: BASCOM Source Code von Peter DH1PAX

'Frequenzeingabe über Tastatur für FT 817

'Peter Horlacher DH1PAX

'13.03.2004

' Compilerparameter für BASCOM

\$baud = 4800

'Baudrate

\$crystal = 4915200

'Quarzfrequenz

' Variablendeklaration

Dim A As Byte , I As Byte , D As Byte , E As Byte , F As Byte

Dim G As Word 'Frequenz, 4 Stellen

Dim S As Byte 'Simplexflag

Dim C(16) As Byte 'Array für Tastaturliste

Dim K(10) As Byte 'Array für Frequenz

Dim Aohne(5) As Byte 'ohne Ablage

Dim Aneg(5) As Byte 'negativer Offset

Dim A2m(5) As Byte 'Ablage 2m (nur FT-817)

Dim A70cm(5) As Byte 'Ablage 70cm (nur FT-817)

Dim A10m(5) As Byte 'Ablage 10m (nur FT-817)

' BASCOM verlangt Deklaration aller Unterprogramme im Kopfteil,

' ihre Definition erfolgt weiter unten

Declare Sub Zero\_freq() 'Frequenz löschen

Declare Sub Check2m() 'Relaisablage 2m prüfen

Declare Sub Check70cm() 'dito, 70cm

Declare Sub Check10m() 'dito, 10m

' Definition der Umwandlungstabelle für die Scancodes der Tastaturmatrix

' in den Zahlenwert der Taste (mittels Arrayvariable C)

C(1) = 1

C(2) = 2

C(3) = 3

Rem c(4) = 'a' 'Bei Matrix 4\*4 vierte Spalte

C(5) = 4

C(6) = 5

C(7) = 6

Rem c(8) = 'b' 'dito, noch unbenutzt

C(9) = 7

C(10) = 8

C(11) = 9

Rem c(12) = 'c' 'dito

C(13) = 12

C(14) = 0

C(15) = 14

Rem c(16) = 'd' 'dito

' Array K initialisieren, mit Frequenz 00 00 00 00:

Call Zero\_freq

' und mit Befehlsword zum Setzen der Frequenz:

K(5) = &H01 'Befehlsword 01 bedeutet Frequenz setzen (FT-817)

'K(5) = &H0A 'anderer Code für FT-100 zum Setzen der Frequenz

' Definition der Arrays mit Befehlsfolgen zur Wahl von Simplexbetrieb  
' und ggf. erforderlicher Shift bei Relaisfrequenzen

```
' simplex
Aohne(1) = &H89          ' FT-817
Aohne(2) = &H00
Aohne(3) = &H00
Aohne(4) = &H00
Aohne(5) = &H09
```

```
'Aohne(1) = &H00        ' FT-100
'Aohne(2) = &H00
'Aohne(3) = &H00
'Aohne(4) = &H00
'Aohne(5) = &H84
```

```
' duplex negativer offset
Aneg(1) = &H09          ' FT-817
Aneg(2) = &H00
Aneg(3) = &H00
Aneg(4) = &H00
Aneg(5) = &H09
```

```
'Aneg(1) = &H00        ' FT-100
'Aneg(2) = &H00
'Aneg(3) = &H00
'Aneg(4) = &H01
'Aneg(5) = &H84
```

'Beim FT-100 kann der Wert des Frequenzversatzes nicht über die CAT-  
'Schnittstelle übertragen werden, auswahl erfolgt im Gerät automatisch.

```
' Ablage Duplex auf 2m: 600kHz
A2m(1) = &H00          ' nur FT-817
A2m(2) = &H06
A2m(3) = &H00
A2m(4) = &H00
A2m(5) = &HF9
```

```
' Ablage Duplex auf 70cm: 7,6 MHz
A70cm(1) = &H00        'nur FT-817
A70cm(2) = &H76
A70cm(3) = &H00
A70cm(4) = &H00
A70cm(5) = &HF9
```

```
' Ablage Duplex auf 10m 100kHz
A10m(1) = &H00         'nur FT-817
A10m(2) = &H01
A10m(3) = &H00
A10m(4) = &H00
A10m(5) = &HF9
```

' hier wird die Unterstützung für Matrixastaturen des BASCOM-Compilers  
' deklariert: Tastatur an Portb mit 100ms delay  
Config Kbd = Portb , Delay = 100

```

I = 0      'Zeichenpuffer ist bei Programmstart noch leer

'Hauptprogramm:

Tast:      'Hauptschleife, wird ab hier endlos durchlaufen

Do         'Tastaturabfrage: wartet auf Tastendruck
A = Getkbd()      'liefert Wert 0-16, blockiert nicht
Loop Until A <> 16    '16 bedeutet keine Taste gedrückt, Warteschleife

Do         'wartet, bis Taste losgelassen wird
D = Getkbd()
Loop Until D = 16

I = I + 1      'Zeichenzähler erhöhen
A = C(a + 1)   'Scancode der Tastaturmatrix nach Ziffer wandeln

If A > 9 Then Goto Ausgabe 'nur Ziffern speichern, sonst überspringen

D = I Mod 2    'wenn i gerade, dann d=0, sonst d=1
E = I \ 2      'ganzzahlige Division ohne Rest, z.B. 3\2 ergibt 1
E = E + D      'Index für Tastaturarray
F=E           'nur für FT-817 verwenden,
'F = 5 - E     'sonst diese Änderung für FT-100

If D = 1 Then  'nur fuer ungerade I
Shift A, Left, 4 'untere 4 Bit in die oberen 4 Bit schieben
K(f) = A And &HF0 'obere 4 Bit maskieren und im Frequenzstring merken
Else          'nur fuer gerade I
A = A And &H0F   'nur untere 4 Bit maskieren
K(f) = K(f) + A  'und zum Frequenzstring dazuaddieren
End If

Ausgabe:

If A = 12 Then 'unbelegte Taste, hier wurde '*' gedrückt
'Programmerweiterungen, etwa Auswertung der '*'-Taste oder einer größeren
'Tastaturmatrix, können ab hier erfolgen.
I = I - 1      'da I schon erhöht wurde, dies zurücknehmen
End If        '*' wirkt also wie gar kein Tastendruck

If I = 8 Or A = 14 Then '8 Stellen eingegeben oder Raute gedrückt

Printbin K(1) ; 5      'Frequenz über serielles Interface im FT-100 setzen
Waitms 200            'kurze Zwangspause
S = 1                 'Flag: Betrieb ohne Ablage als Default merken

'Ggf. Shift ermitteln und setzen (FT-817)
G = K(2)              '6. und 5. Stelle der Frequenz in k(2)
Shift G, Left, 8      'in die oberen 8 Bit der 16 Bit-Variable G schieben
G = G + K(3)          'und 4. und 3. Stelle in die unteren 8 Bit:
'ergibt also eine Variable mit 1MHz-, 100kHz-, 10kHz-, 1kHz-Werten der Frequenz
'für weiteren Vergleich

```

```

Select Case K(1)      'erste beide Stellen der Frequenz in Byte k(1)

'Ggf. Shift ermitteln und setzen (FT-100)
' G = K(3)           '6. und 5. Stelle der Frequenz in k(3)
' Shift G , Left , 8   'in die oberen 8 Bit der 16 Bit-Variable G schieben
' G = G + K(2)       'und 4. und 3. Stelle in die unteren 8 Bit:
'ergibt also eine Variable mit 1MHz-, 100kHz-, 10kHz-, 1kHz-Werten der Frequenz
'für weiteren Vergleich
' Select Case K(4)    'erste beide Stellen der Frequenz in Byte k(4)
  Case &H14 : Call Check2m '2m-Relaisfrequenz? Prüfung in Check2m
  Case &H43 : Call Check70cm '70cm-Relaisfrequenz? Prüfung in Check70cm
  Case &H02 : Call Check10m '10m-Relaisfrequenz? Prüfung in Check10m
End Select

If S = 1 Then        'falls Simplexbetrieb,
  Printbin Aohne(1) ; 5 'diesen auswaehlen
  Waitms 200
End If

Call Zero_freq      'alte Frequenz löschen
I = 0               'es sind also bislang 0 Stellen eingegeben

End If
Goto Tast           'Endlosschleife mittels Rücksprung schließen

```

' Unterprogrammdefinitionen stehen ab hier:

```

Sub Zero_freq()     'die vier Bytes zur Aufnahme der Frequenz auf 00 setzen
  K(1) = 0
  K(2) = 0
  K(3) = 0
  K(4) = 0
End Sub

```

Rem auf 2m 145,600 MHz. bis 145,800 MHz. wird 0,6 MHz. tiefer gesendet

```

Sub Check2m()
  If G >= &H5600 And G <= &H5800 Then    '145,600 MHz - 145,800 MHz
  Printbin A2m(1) ; 5                    'Frequenz -00,600000 MHz
  Waitms 200
  Printbin Aneg(1) ; 5                    'Ablage neg. shift
  Waitms 200
  S = 0                                  'Duplexbetrieb gesetzt Flag
  End If
End Sub

```

Rem auf 70cm 438,650 MHz. bis 439,425 MHz. wird 7,6 MHz. tiefer gesendet

```

Sub Check70cm
  If G >= &H8650 And G < &H9425 Then    '438,650 MHz - 439,425 MHz
  Printbin A70cm(1) ; 5

```



```
Waitms 200
Printbin Aneg(1) ; 5          'Ablage neg. shift
Waitms 200
S = 0                        'Duplexbetrieb gesetzt Flag
End If
End Sub
```

Rem auf 10m 29,660 MHz. bis 29,690 MHz. wird 0,1 MHz. tiefer gesendet

```
Sub Check10m()
If G >= &H9660 And G <= &H9690 Then      '29,660 MHz - 29,690 MHz
Printbin A10m(1) ; 5                    'Frequenz -00,100000 MHz
Waitms 200
Printbin Aneg(1) ; 5                    'Ablage neg. shift
Waitms 200
S = 0                                    'Duplexbetrieb gesetzt Flag
End If
End Sub
```

' E N D E